incorporated to move the object, without rotation, relative to the co-ordinate origin. You could attempt to incorporate a routine to remove the lines and parts of lines that should be hidden from view. This makes the present wire frame image much more realistic. However, such hidden line removal is a vastly complicated matter. It requires very complex mathematics and would slow the program down a great deal. Even if you never add to this program but leave it exactly as it is given here, you can still achieve some pretty spectacular, or just pretty, results.

Spectrum Version

```
 10 LET N=16
 20 DIM P(50)
 21 DIM X(50)
 22 DIM Y(50)
 23 DIM Z(50)
 24 DIM A(50)
 25 DIM B(50)
 40 LET D=10: LET P=0.5
 50 LET SI=SIN 0.09: LET CO=COS 0.09
 60 FOR I=1 TO N
 70 READ P(I),X(I),Y(I),Z(I)
 80 NEXT I
 90:
200 INVERSE 0: GO SUB 300
210 IF INKEY$<>"" THEN  GO TO 210
211 IF INKEY$="" THEN  GO TO 211
212 LET I$=INKEY$
230 INVERSE 1: GO SUB 300
240 IF I$="1" THEN  GO SUB 1000
241 IF I$="2" THEN  GO SUB 2000
242 IF I$="3" THEN  GO SUB 3000
243 IF I$="4" THEN  GO SUB 4000
244 IF I$="5" THEN  GO SUB 5000
245 IF I$="6" THEN  GO SUB 6000
246 IF I$="7" THEN  GO SUB 7000
247 IF I$="8" THEN  GO SUB 8000
250 GO TO 200
260:
300 FOR I=1 TO N
310 LET A(I)=X(I)*300/(P*Z(I)+D): LET B(I)=Y(
I)*300/(P*Z(I)+D)
320 NEXT I
330 FOR I=1 TO N
340 IF P(I)=4 THEN  PLOT A(I)+128,B(I)+85
345 IF P(I)=5 THEN  DRAW A(I)-A(I-1),B(I)-B(I
-1)
350 NEXT I
360 RETURN
370:
1000 FOR I=1 TO N
1010 LET X=X(I)*CO-Z(I)*SI
1020 LET Z=Z(I)*CO+X(I)*SI
1030 LET X(I)=X: LET Z(I)=Z
1040 NEXT I
1050 RETURN
1060:
2000 FOR I=1 TO N
2010 LET X=X(I)*CO+Z(I)*SI
2020 LET Z=Z(I)*CO-X(I)*SI
2030 LET X(I)=X: LET Z(I)=Z
2040 NEXT I
2050 RETURN
2060:
3000 FOR I=1 TO N
3010 LET Y=Y(I)*CO+Z(I)*SI
3020 LET Z=Z(I)*CO-Y(I)*SI
3030 LET Y(I)=Y: LET Z(I)=Z
3040 NEXT I
3050 RETURN
3060:
4000 FOR I=1 TO N
4010 LET Y=Y(I)*CO-Z(I)*SI
4020 LET Z=Z(I)*CO+Y(I)*SI
4030 LET Y(I)=Y: LET Z(I)=Z
4040 NEXT I
4050 RETURN
4060:
5000 LET D=D*0.9
5010 RETURN
5020:
6000 LET D=D/0.9
6010 RETURN
6020:
7000 LET P=P/0.9
7010 RETURN
7020:
8000 LET P=P*0.9
8010 RETURN
8020:
9000 DATA 4,1,1,1, 5,1,1,-1, 5,-1,1,-1, 5,-1,1
,1,5,1,1,1
9010 DATA 5,1,-1,1, 5,1,-1,-1, 5,-1,-1,-1, 5,-
1,-1,1, 5,1,-1,1
9020 DATA 4,1,-1,-1, 5,1,1,-1, 5,-1,-1,-1, 5,-
1,1,-1
9030 DATA 4,-1,1,1, 5,-1,-1,1
```

BBC Version

```
  10 N=16
  20 DIM P(50),X(50),Y(50),Z(50),A(50),B(50)
  30 MODE0:VDU29,640;512;5
  40 D=10:P=0.5
  50 SI=SIN(0.09):CO=COS(0.09)
  60 FOR I=1 TO N
  70   READ P(I),X(I),Y(I),Z(I)
  80   NEXT I
  90 :
 200 GCOL0,3:GOSUB 300
 210 I$=GET$
 220 V=VAL(I$)
 230 GCOL0,0:GOSUB 300
 240 ON V GOSUB 1000,2000,3000,4000,5000,6000
,7000,8000 ELSE 150
 250 GOTO 200
 260 :
 300 FOR I=1TON
 310   A(I)=X(I)*1000/(P*Z(I)+D):B(I)=Y(I)*
1000/(P*Z(I)+D)
 320   NEXT I
 330 FOR I=1 TO N
 340   PLOT P(I),A(I),B(I)
 350   NEXT I
 360 RETURN
 370 :
1000 FOR I=1 TO N
1010   X=X(I)*CO-Z(I)*SI
1020   Z=Z(I)*CO+X(I)*SI
1030   X(I)=X:Z(I)=Z
1040   NEXT I
1050 RETURN
1060 :
2000 FOR I=1 TO N
2010   X=X(I)*CO+Z(I)*SI
2020   Z=Z(I)*CO-X(I)*SI
2030   X(I)=X:Z(I)=Z
2040   NEXT I
2050 RETURN
2060 :
3000 FOR I=1 TO N
3010   Y=Y(I)*CO+Z(I)*SI
3020   Z=Z(I)*CO-Y(I)*SI
3030   Y(I)=Y:Z(I)=Z
3040   NEXT I
3050 RETURN
3060 :
4000 FOR I=1 TO N
4010   Y=Y(I)*CO-Z(I)*SI
4020   Z=Z(I)*CO+Y(I)*SI
4030   Y(I)=Y:Z(I)=Z
4040   NEXT I
4050 RETURN
4060 :
5000 D=D*.9
5010 RETURN
5020 :
6000 D=D/.9
6010 RETURN
6020 :
7000 P=P/.9
7010 RETURN
7020 :
8000 P=P*.9
8010 RETURN
8020 :
10000 DATA 4,1,1,1, 5,1,1,-1, 5,-1,1,-1, 5,-1,
1,1,5,1,1,1
10010 DATA 5,1,-1,1, 5,1,-1,-1, 5,-1,-1,-1, 5,
-1,-1,1, 5,1,-1,1
10020 DATA 4,1,-1,-1, 5,1,1,-1, 5,-1,-1,-1, 5,
-1,1,-1
10030 DATA 4,-1,1,1, 5,-1,-1,1
```



KEVIN JONES